
AshPhoenix
Welcome! This is the package for integrating Phoenix Framework and Ash Framework. It provides tools for integrating with Phoenix forms (AshPhoenix.Form), Phoenix LiveViews (AshPhoenix.LiveView) and more.

  
    
  
  Installation


Add ash_phoenix to your list of dependencies in mix.exs:
{:ash_phoenix, "~> 2.1.14"}

  
    
  
  Whats in the box?


	AshPhoenix.Form - A form data structure for using resource actions with phoenix forms
	AshPhoenix.Form.Auto - Tools to automatically determine nested form structures based on calls to manage_relationship for an action.
	AshPhoenix.FilterForm - A form data structure for building filter statements
	AshPhoenix.LiveView - Helpers for querying data and integrating changes
	AshPhoenix.SubdomainPlug - A plug to determine a tenant using subdomains for multitenancy
	AshPhoenix.FormData.Error - A protocol to allow errors to be rendered in forms
	Phoenix.HTML.Safe implementations for Ash.CiString, Ash.NotLoaded and Decimal
	mix ash_phoenix.gen.live for generating liveview modules
	mix ash_phoenix.gen.html for generating controllers and views


  
    
  
  Tutorials


	Getting Started with Ash and Phoenix


  
    
  
  Topics


	Union Forms
	Nested Forms



  

    
Get Started with Ash and Phoenix
    


  
    
  
  Goals


In this guide we will:
	Create a new Phoenix project
	Setup Ash, AshPhoenix and AshPostgres as dependencies
	Create a basic Blog.Post resource
	Create and migrate the database
	Learn how to interact with your resource
	Integrate a minimal Phoenix LiveView with Ash


  
    
  
  Preparation


	Install Elixir
	Phoenix - Up and Running Guide
	Design Principles


  
    
  
  Requirements


If you want to follow along yourself, you will need the following things:
	Elixir (1.12 or later) and Erlang (22 or later) installed
	PostgreSQL installed
	A text editor
	A terminal to run the examples


  
    
  
  Setup



  
    
  
  Create a New Phoenix Project



  
    
  
  Install Phoenix


This section is based on the Phoenix installation docs. For more details go there.
First we need to install the Phoenix project generator, then we'll run the generator to create our new project.
# install Phoenix project generator
$ mix archive.install hex phx_new

# generate Phoenix project
$ mix igniter.new my_ash_phoenix_app --install ash,ash_phoenix,ash_postgres --with phx.new

# cd into project
$ cd my_ash_phoenix_app


  
    
  
  Don't run mix ecto.create


Do not run mix ecto.create, (as it asks you to) we will do this the Ash way later.

  
    
  
  Edit Config


We need to specify the Ash domains that our application uses.
Add this to your config:
# config/config.exs

import Config

config :my_ash_phoenix_app,
  ash_domains: [MyAshPhoenixApp.Blog]

  
    
  
  Create the Domain and add Resources


An Ash domain can be thought of as a Bounded Context in Domain Driven Design terms and can seen as analogous to a Phoenix context. Put simply, its a way of grouping related resources together. In our case our domain will be called MyAshPhoenixApp.Blog.
An Ash domain points to Ash resources. An Ash domain can point to one or more resources. In our case we will only have a single resource MyAshPhoenixApp.Blog.Post. We'll be taking a deeper look into that in the next section.
For now take a look at the Blog domain and the associated resources:
# lib/my_ash_phoenix_app/blog/blog.ex

defmodule MyAshPhoenixApp.Blog do
  use Ash.Domain

  resources do
    resource MyAshPhoenixApp.Blog.Post do
      # Define an interface for calling resource actions.
      define :create_post, action: :create
      define :list_posts, action: :read
      define :destroy_post, action: :destroy
      define :get_post, args: [:id], action: :by_id
    end
  end
end

  
    
  
  Creating Resources



  
    
  
  Creating the Post Resource


A resource is a central concept in Ash. In short, a resource is a domain model object in your system. A resource defines the data it holds and defines the actions that can operate on that data.
When we create Post we will place it in lib/my_ash_phoenix_app/blog/post.ex. So the structure after making the resource should look like so:
lib/
├─ my_ash_phoenix_app/
│  ├─ blog/
│  │  ├─ blog.ex
│  │  ├─ post.ex
Below is the resource module. Read the comments carefully, every line is explained:
# lib/my_ash_phoenix_app/blog/post.ex

defmodule MyAshPhoenixApp.Blog.Post do
  # Using Ash.Resource turns this module into an Ash resource.
  use Ash.Resource,
    # Tells Ash where the generated code interface belongs
    domain: MyAshPhoenixApp.Blog,
    # Tells Ash you want this resource to store its data in Postgres.
    data_layer: AshPostgres.DataLayer

  # The Postgres keyword is specific to the AshPostgres module.
  postgres do
    # Tells Postgres what to call the table
    table "posts"
    # Tells Ash how to interface with the Postgres table
    repo MyAshPhoenixApp.Repo
  end

  actions do
    # Exposes default built in actions to manage the resource
    defaults [:read, :destroy]

    create :create do
      primary? true
      # accept title as input
      accept [:title]
    end

    update :update do
      primary? true
      # accept content as input
      accept [:content]
    end

    # Defines custom read action which fetches post by id.
    read :by_id do
      # This action has one argument :id of type :uuid
      argument :id, :uuid, allow_nil?: false
      # Tells us we expect this action to return a single result
      get? true
      # Filters the `:id` given in the argument
      # against the `id` of each element in the resource
      filter expr(id == ^arg(:id))
    end
  end

  # Attributes are simple pieces of data that exist in your resource
  attributes do
    # Add an autogenerated UUID primary key called `:id`.
    uuid_primary_key :id
    # Add a string type attribute called `:title`
    attribute :title, :string do
      # We don't want the title to ever be `nil`
      allow_nil? false
    end

    # Add a string type attribute called `:content`
    # If allow_nil? is not specified, then content can be nil
    attribute :content, :string
  end
end

  
    
  
  Creating and Migrating the Database


We have specified the resource in Ash. But we have yet to create it in our data layer (in our case Postgres).
First we need to create our database:
$ mix ash.setup

Running setup for AshPostgres.DataLayer...
The database for MyAshPhoenixApp.Repo has been created

01:23:45.678 [info] Migrations already up

Now we need to populate our database. We do this by generating and performing a migration.
We can use a generator to produce a migration for us. Ash can deduce what needs to go into the migration and do the hard work for us, to do this use the command below:
$ mix ash.codegen initial_migration

# ... don't worry about other files it creates

Generating Migrations:
* creating priv/repo/migrations/20230208045101_initial_migration.exs

Here is the migration file commented in detail:
# priv/repo/migrations/20230208045101_initial_migration.exs

defmodule MyAshPhoenixApp.Repo.Migrations.InitialMigration do
  use Ecto.Migration

  # This function runs when migrating forward
  def up do
    # Creates the `:posts` table
    create table(:posts, primary_key: false) do
      # Adds primary key attribute `:id` of type `:uuid`
      # null values are not allowed
      add :id, :uuid, null: false, default: fragment("gen_random_uuid()"), primary_key: true

      # Adds attribute `:title` of type `:text`, null values are not allowed
      add :title, :text, null: false
      # Adds attribute `:content` of type `:text`, null values are allowed
      add :content, :text
    end
  end

  # This is the function that runs if you want to rollback the migration.
  def down do
    # Deletes the `:posts` table
    drop table(:posts)
  end
end
We can run the up/0 function which will perform the desired operations on the Postgres database. We do this with the migrate command:
$ mix ash.migrate

In case you want to drop the database and start over again during development you can use mix ash.reset.


  
    
  
  Interacting with your Resources


All interaction with your resource attributes always occur through an action. In our resource we are using the default actions for :create, :read, :update, :destroy along with a custom action :by_id.
:create and :update and :destroy actions require a changeset. Ash changesets are conceptually similar to Ecto changesets. They're data structures which represent an intended change to an Ash resource and provide validation.
The :read action takes a query instead of a changeset.
Below is the most verbose way of calling your resource. All other ways of interaction are some kind of shorthand of these. This means at some point a changeset is being created and passed to the domain, even if it's encapsulated within another function.
# create post
new_post =
  MyAshPhoenixApp.Blog.Post
  |> Ash.Changeset.for_create(:create, %{title: "hello world"})
  |> Ash.create!()

# read all posts
MyAshPhoenixApp.Blog.Post
|> Ash.Query.for_read(:read)
|> Ash.read!()

# get single post by id
MyAshPhoenixApp.Blog.Post
|> Ash.Query.for_read(:by_id, %{id: new_post.id})
|> Ash.read_one!()

# update post
updated_post =
  new_post
  |> Ash.Changeset.for_update(:update, %{content: "hello to you too!"})
  |> Ash.update!()

# delete post
new_post
|> Ash.Changeset.for_destroy(:destroy)
|> Ash.destroy!()
As stated above, this is verbose so Ash has a built in shortcut - The code_interface. You may notice this has already been done in your Post resource inside of the domain module.

  
    
  
  you can call code interfaces whatever you like


The function name doesn't have to match the action name in any way. You could also write:
define :make_post, action: :create
That's perfectly valid and could be called via Blog.make_post/2.
Now we can call our resource like so:
# create post
new_post = MyAshPhoenixApp.Blog.create_post!(%{title: "hello world"})

# read post
MyAshPhoenixApp.Blog.list_posts!()

# get post by id
MyAshPhoenixApp.Blog.get_post!(new_post.id)

# update post
updated_post = MyAshPhoenixApp.Blog.update_post!(new_post, %{content: "hello to you too!"})

# delete post
MyAshPhoenixApp.Blog.destroy_post!(updated_post)
Now isn't that more convenient?

  
    
  
  raising and non-raising functions


All functions that interact with an Ash resource have a raising and non-raising version. For example there are two create functions create/2 and create!/2. create/2 returns {:ok, resource} or {:error, reason}. create!/2 will return just the record on success and will raise an error on failure.

  
    
  
  Connecting your Resource to a Phoenix LiveView


Now we know how to interact with our resource, we can generate a liveview for it!
let's run mix ash_phoenix.gen.live to generate a liveview! Run the following command,
declining to name your actor, accepting any other default values, and following the 
instructions listed at the end for adding the liveview to your router.
mix ash_phoenix.gen.live --domain MyAshPhoenixApp.Blog --resource MyAshPhoenixApp.Blog.Post --resourceplural posts

Now, start the web server by running mix phx.server. Then, visit the posts route that you added in your browser to see what we have just created.
You can see how using functions created by our code_interface makes it easy to integrate Ash with Phoenix.
You may also notice this is the first time we've used the AshPhoenix library. The AshPhoenix library contains utilities to help Ash integrate with Phoenix and LiveView Seamlessly. One of these utilities is AshPhoenix.Form which can automatically produce changesets to be used in the forms.
That's it for this guide. We've gone from 0 to a fully working Phoenix App using Ash. To get a closer look, see the accompanying repo here.

  
    
  
  Where to Next?


We are really just scratching the surface of what can be done in Ash. Look below for what to look at next.

  
    
  
  Continue Learning


There's a few places you can go to learn more about how to use ash:
	Read more about how to query the data in your resources - Ash.Query
	Dig deeper into actions.
	Study resource relationship management


  
    
  
  Ash Authentication & Ash Authentication Phoenix


See the power Ash can bring to your web app or API. Get authentication working in minutes.

  
    
  
  Add an API (or two)


Check out the AshJsonApi and AshGraphql extensions to effortlessly build APIs around your resources.


  

    
Union Forms
    

When building a form for a union, you use inputs_for as normal, but a few things are done for you under the hood.
Lets take this example union:
defmodule NormalContent do
  use Ash.Resource, data_layer: :embedded

  attributes do
    attribute :body, :string, allow_nil?: false
  end

  actions do
    defaults [:read, create: [:body], update: [:body]]
  end
end

defmodule SpecialContent do
  use Ash.Resource, data_layer: :embedded

  attributes do
    attribute :text, :string, allow_nil?: false
  end

  actions do
    defaults [:read, create: [:text], update: [:text]]
  end
end

defmodule Content do
  use Ash.Type.NewType,
    subtype_of: :union,
    constraints: [
      types: [
        normal: [
          type: NormalContent,
          tag: :type,
          tag_value: :normal
        ],
        special: [
          type: SpecialContent,
          tag: :type,
          tag_value: :special
        ]
      ]
    ]
end

  
    
  
  Determining the type for a union form


We track the type of the value in a hidden param called _union_type. You can use this to show a different form depending on the type of thing.

  
    
  
  Changing the type of a union form


If you want to let the user change the union type, you would use AshPhoenix.Form.remove_form/3 and AshPhoenix.Form.add_form/3. See the example below for the template, and here is an example event handler
def handle_event("type-changed", %{"_target" => path} = params, socket) do
  new_type = get_in(params, path)
  # The last part of the path in this case is the field name
  path = :lists.droplast(path)

  form =
    socket.assigns.form
    |> AshPhoenix.Form.remove_form(path)
    |> AshPhoenix.Form.add_form(path, params: %{"_union_type" => new_type})

  {:noreply, assign(socket, :form, form)}
end

  
    
  
  Non-embedded types


If one of your union values is a non embedded type, like :integer, it will still be a nested form, but you would access the single value with <.input field={nested_form[:value]} .../>

  
    
  
  Example


We might have a form like this:
<.inputs_for :let={fc} field={@form[:content]}>
  <!-- Dropdown for setting the union type -->
  <.input
    field={fc[:_union_type]}
    phx-change="type-changed"
    type="select"
    options={[Normal: "normal", Special: "special"]}
  />

  <!-- switch on the union type to display a form -->
  <%= case fc.params["_union_type"] do %>
    <% "normal" -> %>
      <.input  type="text" field={fc[:body]} />
    <% "special" -> %>
      <.input type="text" field={fc[:text]} />
  <% end %>
</.inputs_for>


  

    
Nested Forms
    

Make sure you're familiar with the basics of AshPhoenix.Form before reading this guide.
When we talk about "nested" or "related" forms, we mean sets of form inputs
that are for resource actions for related or embedded resources.
For example, you might have a form for creating a "business" that can also
include multiple "locations". In some cases, you may have buttons to add or 
remove from a list of nested forms, you may be able to drag and drop to reorder
forms, etc. In other cases, the form may just be for one related thing, think
a form for updating a "user" that also contains inputs for its associated "profile".

  
    
  
  Defining the structure



  
    
  
  Inferring from the action


AshPhoenix.Form automatically infers what "nested forms" are available, based on introspecting actions
which use change manage_relationship. For example, in the following action:
# on a `MyApp.Operations.Business` resource
create :create do
  accept [:name]

  argument :locations, {:array, :map}

  change manage_relationship(:locations, type: :create)
end
With this action, you could submit an input like so:
%{name: "Wally World", locations: [%{name: "HQ", address: "1 hq street"}]}
AshPhoenix.Form will look at the action, allowing you to use Phoenix's
<.inputs_for component for locations. Here is what it might look like in
practice:
<.simple_form for={@form} phx-change="validate" phx-submit="submit">
  <.input field={@form[:email]} />

  <.inputs_for :let={location} field={@form[:locations]}>
    <.input field={location[:name]} />
  </.inputs_for>
</.form>
To turn this automatic behavior off, you can specify forms: [auto?: false] 
when creating the form.

  
    
  
  Manually defining nested forms


You can manually specify nested form configurations using the forms option.
For example:
AshPhoenix.Form.for_create(
  MyApp.Operations.Business, 
  :create, 
  forms: [
    locations: [
      type: :list,
      resource: MyApp.Operations.Location,
      create_action: :create
    ]
  ]
)
You should prefer to use the automatic form definition wherever possible,
but this exists as an escape hatch to customize configuration.

  
    
  
  Updating existing data


You should be sure to load any relationships that are necessary for your
manage_relationships when you want to update the nested items. 
For example, if the form above was for an update action,
you may want to allow updating the existing locations all in a single form.
AshPhoenix.Form will show a form for each existing location, but only if
the locations are loaded on the business already. For example:
business = Ash.load!(business, :locations)

form = AshPhoenix.Form.for_update(business, :update)

  
    
  
  Not using tailwind?


If you're not using tailwind, you'll need to replace class="hidden"
in the examples below with something else. In standard HTML, you'd do
<input .... hidden />. As long as the checkbox is hidden, you're good!

  
    
  
  Adding nested forms


There are two ways to add nested forms.

  
    
  
  The _add_* checkbox


<.simple_form for={@form} phx-change="validate" phx-submit="submit">
  <.input field={@form[:email]} />

  <.inputs_for :let={location} field={@form[:locations]}>
    <.input field={location[:name]} />
  </.inputs_for>

  <label>
    <input
      type="checkbox"
      name={"#{@form.name}[_add_locations]"}
      value="end"
      class="hidden"
    />
    <.icon name="hero-plus" />
  </label>
</.form>
This checkbox, when checked, will add a parameter like form[_add_locations]=end. 
When AshPhoenix.Form is handling nested forms, it will see that and append an empty
form at the end. Valid values are "start", "end" and an index, i.e "3", in which
case the new form will be inserted at that index.

  
    
  
  But the checkbox is hidden, what gives?


If you're anything like me, the label + checkbox combo above may confuse you 
at first sight. When you have a checkbox inside of a label, clicking on the label
counts as clicking the checkbox itself!

  
    
  
  AshPhoenix.Form.add_form


In some cases, you may want to add a form either in a way that can't be triggered by a checkbox
or that requires some additional data (like non-empty starting params). In those cases,
you can use a button and a handle_event For example:
<.simple_form for={@form} phx-change="validate" phx-submit="submit">
  <.input field={@form[:email]} />

  <.inputs_for :let={location} field={@form[:locations]}>
    <.input field={location[:name]} />
  </.inputs_for>

  <.button type="button" phx-click="add-form" phx-value-path={@form.name <> "[locations]"}>
    <.icon name="hero-plus" />
  </.button>
</.form>

  
    
  
  whats with @form.name <> "[locations]"


By always using a path "relative" to the root form, we can handle cases where we are
adding a form to a multiply-nested form. So the path could be somethign like
locations[0][addresses][1]. The event handler has to know exactly where we are adding
a form. In the example above, we could just say add_form(form, :locations). It would
be simpler, but we want to highlight how to work with potentially deeply nested data.
def handle_event("add-form", %{"path" => path}, socket) do
  form = AshPhoenix.Form.add_form(socket.assigns.form, path, params: %{
    address: "Put your address here!"
  })

  {:noreply, assign(socket, :form, form)}
end

  
    
  
  Removing nested forms


Just like adding nested forms, there are two ways to remove nested forms.

  
    
  
  Using the _drop_* checkbox


The _drop_* checkbox uses checkboxes which add form indices to a list that should
be removed from the list. For example, given the following:
<.simple_form for={@form} phx-change="validate" phx-submit="submit">
  <.input field={@form[:email]} />

  <.inputs_for :let={location} field={@form[:locations]}>
    <.input field={location[:name]} />

    <label>
      <input
        type="checkbox"
        name={"#{@form.name}[_drop_locations][]"}
        value={location_form.index}
        class="hidden"
      />

      <.icon name="hero-x-mark" />
    </label>
  </.inputs_for>
</.form>
When the checkbox is checked, the server sees:
%{"form" => %{"_drop_locations" => ["0"]}}
We use this information to automatically remove the item at that index on validate.

  
    
  
  Using AshPhoenix.Form.remove_form


Just like adding forms, there is a manual way to remove forms. In this case
we pass the full path to the form being removed.
<.simple_form for={@form} phx-change="validate" phx-submit="submit">
  <.input field={@form[:email]} />

  <.inputs_for :let={location} field={@form[:locations]}>
    <.input field={location[:name]} />

    <.button type="button" phx-click="remove-form" phx-value-path={location.name}>
      <.icon name="hero-x-mark" />
    </.button>
  </.inputs_for>
</.form>
def handle_event("remove-form", %{"path" => path}, socket) do
  form = AshPhoenix.Form.remove_form(socket.assigns.form, path)

  {:noreply, assign(socket, :form, form)}
end

  
    
  
  Sorting nested forms


Just like adding and removing forms, there are two ways to sort nested forms too!

  
    
  
  Using _sort_* checkboxes


This method is useful when combined with something like sortable.js 
to allow for dragging and dropping on the front end.

  
    
  
  the order_is_key option


If you are working with a sorted relationship, you will likely want to couple it
with the order_is_key option of managed_relationships. This writes the order
of items in the list of inputs into each input, as if it was provided as an input
change manage_relationship(:locations, type: :direct_control, order_is_key: :position)
In the above example, if you provided a list of inputs like
[%{address: "foo"}, %{address: "bar"}], it would first be converted into
[%{address: "foo, order: 0}, %{address: "bar", order: 1}] before being
processed.
Lets say you had the following Sortable hook in your app.js
import Sortable from "sortablejs"

export const Sortable = {
  mounted() {
    new Sortable(this.el, {
      animation: 150,
      draggable: '[data-sortable="true"]',
      ghostClass: "bg-yellow-100",
      dragClass: "shadow-2xl",
      onEnd: (evt) => {
        this.el.closest("form").querySelector("input").dispatchEvent(new Event("input", {bubbles: true}))
      }
    })
  }
}
...

let Hooks = {}

Hooks.Sortable = Sortable
You could use the _sort_* checkbox in each nested form like so:
<.simple_form for={@form} phx-change="validate" phx-submit="submit">
  <.input field={@form[:email]} />

  <div id="location-list" phx-hook="Sortable">
    <.inputs_for :let={location} field={@form[:locations]}>
      <div data-sortable="true">
        <input
          type="hidden"
          name={"#{@form.name}[_sort_locations][]"}
          value={location_form.index}
        />

        <.input field={location[:name]} />
      </div>
    </.inputs_for>
</.form>
In this case you'd drag the entire div. sortable.js supports all kinds of useful features,
like drag handles. See their docs for more. 
Now, lets say you were to drag the second form above the first form, the server would see the
params as:

%{"form" => %{"_sort_locations" => ["1", "0"]}}
AshPhoenix.Form would then sort the nested forms accordingly.

  
    
  
  Using AshPhoenix.Form.sort_forms/3


The manual way is using  AshPhoenix.Form.sort_forms/3. This can be used
to move a specific element up or down, or to sort all forms. sortable.js
can be used in such a way that it provides the full sorting back to your
server. 
Providing a full sort order
This could be used to send a handle_event that gives you a list
of indices in a new order. An example of that setup can be seen 
here. Keep in mind that you'll want to adjust the method to extract a field from
each element of the current index, using something like data-current-index={location_form.index} to
store the index.
indices might look something like this: ["0", "1", "3", "2"]
def handle_event("update-sorting", %{"path" => path, "indices" => indices}, socket) do
  form = AshPhoenix.Form.sort_forms(socket, path, indices)
  {:noreply, assign(socket, form: form)}
end
Moving a specific form up
If you wanted up/down buttons, you could use event handlers like the following.
def handle_event("move-up", %{"path" => form_to_move}, socket) do
  # decrement typically means "move up" visually
  # because forms are rendered down the page ascending
  form = AshPhoenix.Form.sort_forms(socket, form_to_move, :decrement)
  {:noreply, assign(socket, form: form)}
end

def handle_event("move-down", %{"path" => form_to_move}, socket) do
  # increment typically means "move down" visually
  # because forms are rendered down the page ascending
  form = AshPhoenix.Form.sort_forms(socket, form_to_move, :increment)
  {:noreply, assign(socket, form: form)}
end

  
    
  
  Putting it all together


Lets look at what it looks like with all of the checkbox-based features in one:
defmodule MyApp.MyForm do
  use MyAppWeb, :live_view

  def render(assigns) do
    ~H"""
    <.simple_form for={@form} phx-change="validate" phx-submit="submit">
      <.input field={@form[:email]} />

      <!-- Use sortable.js to allow sorting nested input -->
      <div id="location-list" phx-hook="Sortable">
        <.inputs_for :let={location} field={@form[:locations]}>
          <!-- inputs each nested location -->
          <div data-sortable="true">
            <!-- AshPhoenix.Form automatically applies this sort -->
            <input
              type="hidden"
              name={"#{@form.name}[_sort_locations][]"}
              value={location_form.index}
            />

            <.input field={location[:name]} />

            <!-- AshPhoenix.Form automatically removes items when checked -->
            <label>
              <input
                type="checkbox"
                name={"#{@form.name}[_drop_locations][]"}
                value={location_form.index}
                class="hidden"
              />

              <.icon name="hero-x-mark" />
            </label>
          </div>
        </.inputs_for>

        <!-- AshPhoenix.Form automatically appends a new item when checked -->
        <label>
          <input
            type="checkbox"
            name={"#{@form.name}[_add_locations]"}
            value="end"
            class="hidden"
          />
          <.icon name="hero-plus" />
        </label>
      </div>
    </.form>
    """
  end

  def mount(_params, _session, socket) do
    {:ok, assign(socket, form: MyApp.Operations.form_to_create_business())}
  end

  def handle_event(socket, "validate", %{"form" => params}) do
    {:noreply, assign(socket, :form, AshPhoenix.Form.validate(socket.assigns.form, params))}
  end

  def handle_event(socket, "submit", %{"form" => params}) do
    case AshPhoenix.Form.submit(socket.assigns.form, params: params) do
      {:ok, business} ->
        socket =
          socket
          |> put_flash(:success, "Business created successfully")
          |> push_navigate(to: ~p"/businesses/#{business.id}")

        {:noreply, socket}

      {:error, form} ->
        {:noreply, assign(socket, :form, form)}
    end
  end
end


  

    
Change Log
    

All notable changes to this project will be documented in this file.
See Conventional Commits for commit guidelines.

  
    
  
  v2.1.14 (2025-01-19)



  
    
  
  Bug Fixes:


	print routes on ash_phoenix.gen.live again

	properly find matching forms by primary key

	Allow re-adding forms to a nested form after deleting the last one from a list (#291)

	handle case where last form is deleted

	simplifications and fixes for drop_param

	ensure that form interfaces properly set data

	fix warning in filter_form.ex (#285)



  
    
  
  Improvements:


	support _drop_*, _add_* and _sort_* params

	add AshPhoenix.Form.sort_forms utility



  
    
  
  v2.1.13 (2025-01-03)



  
    
  
  Bug Fixes:


	ensure that form interfaces properly set data

	update html generators to properly call actions



  
    
  
  v2.1.12 (2024-12-22)



  
    
  
  Improvements:


	Add AshPhoenix extension


  
    
  
  v2.1.11 (2024-12-20)



  
    
  
  Bug Fixes:


	only ever raise error classes


  
    
  
  Improvements:


	make igniter optional

	simplify setting valid on AshPhoenix.Form.add_error/3

	don't populate args that aren't set



  
    
  
  v2.1.10 (2024-12-12)



  
    
  
  Bug Fixes:


	use Igniter.Project.Module.parse to get module names for generator (#274)


  
    
  
  v2.1.9 (2024-12-11)



  
    
  
  Bug Fixes:


	ensure that errors on before_action hooks invalidate the form


  
    
  
  Improvements:


	Migrate phoenix gen to igniter (#261)

	add AshPhoenix.Form.update_params/2



  
    
  
  v2.1.8 (2024-10-29)



  
    
  
  Improvements:


	track raw_params


  
    
  
  v2.1.7 (2024-10-29)



  
    
  
  Bug Fixes:


	set _union_type param when unnesting a resource in a union

	don't wrap resources inside of unions as WrappedValue

	warn on missing params on submit

	unwrap unions & wrapped values when fetching values



  
    
  
  v2.1.6 (2024-10-17)



  
    
  
  Improvements:


	allow phoenix_live_view rc


  
    
  
  v2.1.5 (2024-10-14)



  
    
  
  Improvements:


	support generic actions (#250)


  
    
  
  v2.1.4 (2024-09-30)



  
    
  
  Bug Fixes:


	properly include calc args in to_filter_map


  
    
  
  v2.1.3 (2024-09-30)



  
    
  
  Bug Fixes:


	properly apply calculations with arguments in filter form


  
    
  
  v2.1.2 (2024-09-03)



  
    
  
  Bug Fixes:


	spec update_form to accept functions of lists


  
    
  
  v2.1.1 (2024-08-01)



  
    
  
  Bug Fixes:


	Use :public? instead of :private? (#221)


  
    
  
  Improvements:


	raise an error on usage of old option name


  
    
  
  v2.1.0 (2024-07-26)



  
    
  
  Bug Fixes:


	ensure we prepare_source for all read action forms


  
    
  
  v2.0.4 (2024-06-13)



  
    
  
  Bug Fixes:


	various fixes for union form handling

	properly fill union values on update

	ensure changing union type is reflected in param transformer



  
    
  
  Improvements:


	honor _union_type when applying input


  
    
  
  v2.0.3 (2024-06-05)



  
    
  
  Bug Fixes:


	properly (i.e safely) encode ci strings to iodata and params

	various union param handling fixes

	properly transform nested params

	make sure that _index is correctly updated before and after removal for sparse forms (#196) (#197)



  
    
  
  v2.0.2 (2024-05-22)



  
    
  
  Bug Fixes:


	don't assume all embeds have a create/update action


  
    
  
  v2.0.1 (2024-05-17)



  
    
  
  Bug Fixes:


	improve union handling

	Convert entered action names into atoms for lookup in the resource (#187)

	various fixes around union forms



  
    
  
  Improvements:


	support adding a form by inserting to an index


  
    
  
  v2.0.0 (2024-04-30)


The changelog is being restarted. See /documentation/1.0-CHANGELOG.md in GitHub for the old changelog.

  
    
  
  Improvements:


	[AshPhoenix.Form] better error message with hints for accepted/non accepted missing forms


  
    
  
  Bug Fixes:


	[AshPhoenix.Form] don't use public_attributes?, check for all accepted attributes. In Ash 3.0, private attributes can be accepted

	[AshPhoenix.Form]

	Pass the tenant to Ash.can/3 and Ash.can?/3. (#165)

	Pass the tenant to Ash.can/3 and Ash.can?/3.




  

    
DSL: AshPhoenix
    

An extension to add form builders to the code interface.
There is currently no DSL for this extension.
This defines a form_to_<name> function for each code interface
function. Positional arguments are ignored, given that in forms,
all input typically comes from the params map.
The generated function passes all options through to
AshPhoenix.Form.for_action/3
Update and destroy actions take the record being updated/destroyed
as the first argument.
For example, given this code interface definition on a domain
called MyApp.Accounts:
resources do
  resource MyApp.Accounts.User do
    define :register_with_password, args: [:email, :password]
    define :update_user, action: :update, args: [:email, :password]
  end
end
Adding the AshPhoenix extension would define 
form_to_register_with_password/2.

  
    
  
  Usage


Without options:
MyApp.Accounts.form_to_register_with_password()
#=> %AshPhoenix.Form{}
With options:
MyApp.Accounts.form_to_register_with_password(params: %{"email" => "placeholder@email"})
#=> %AshPhoenix.Form{}
With 
MyApp.Accounts.form_to_update_user(params: %{"email" => "placeholder@email"})
#=> %AshPhoenix.Form{}


  

    
AshPhoenix 
    



      
An extension to add form builders to the code interface.
There is currently no DSL for this extension.
This defines a form_to_<name> function for each code interface
function. Positional arguments are ignored, given that in forms,
all input typically comes from the params map.
The generated function passes all options through to
AshPhoenix.Form.for_action/3
Update and destroy actions take the record being updated/destroyed
as the first argument.
For example, given this code interface definition on a domain
called MyApp.Accounts:
resources do
  resource MyApp.Accounts.User do
    define :register_with_password, args: [:email, :password]
    define :update_user, action: :update, args: [:email, :password]
  end
end
Adding the AshPhoenix extension would define 
form_to_register_with_password/2.

  
    
  
  Usage


Without options:
MyApp.Accounts.form_to_register_with_password()
#=> %AshPhoenix.Form{}
With options:
MyApp.Accounts.form_to_register_with_password(params: %{"email" => "placeholder@email"})
#=> %AshPhoenix.Form{}
With 
MyApp.Accounts.form_to_update_user(params: %{"email" => "placeholder@email"})
#=> %AshPhoenix.Form{}

      




  

    
AshPhoenix.LiveView 
    



      
Utilities for keeping Ash query results up to date in a LiveView.

      


      
        Summary


  
    Types
  


    
      
        
  
    
    AshPhoenix.SubdomainPlug - ash_phoenix v2.1.14
    
    

    


  
  

    
AshPhoenix.SubdomainPlug 
    



      
This is a basic plug that loads the current tenant assign from a given
value set on subdomain.
This was copied from Triplex.SubdomainPlug, here:
  https://github.com/ateliware/triplex/blob/master/lib/triplex/plugs/subdomain_plug.ex
Options:
	:endpoint (atom/0) - Required. The endpoint that the plug is in, used for deterining the host

	:assign (atom/0) - The key to use when assigning the current tenant The default value is :current_tenant.

	:handle_subdomain - An mfa to call with the conn and a subdomain value. Can be used to do something like fetch the current user given the tenant. Must return the new conn.


To plug it on your router, you can use:
plug AshPhoenix.SubdomainPlug,
  endpoint: MyApp.Endpoint
An additional helper here can be used for determining the host in your liveview, and/or using
the host that was already assigned to the conn.
For example:
def handle_params(params, uri, socket) do
  socket =
    assign_new(socket, :current_tenant, fn ->
      AshPhoenix.SubdomainPlug.live_tenant(socket, uri)
    end)

  socket =
    assign_new(socket, :current_organization, fn ->
      if socket.assigns[:current_tenant] do
        MyApp.Accounts.Ash.get!(MyApp.Accounts.Organization,
          subdomain: socket.assigns[:current_tenant]
        )
      end
    end)

  {:noreply, socket}
end

AshPhoenix.Form 
    



      
A module to allow you to fluidly use resources with Phoenix forms.

  
    
  
  Life cycle


The general workflow is, with either LiveView or Phoenix forms:
	Create a form with AshPhoenix.Form
	Render the form with Phoenix.Component.form (or CoreComponents.simple_form), or, if using Surface, <Form>
	To validate the form (e.g with phx-change for liveview), pass the submitted params to AshPhoenix.Form.validate/3
	On form submission, pass the params to AshPhoenix.Form.submit/2
	On success, use the result to redirect or assign. On failure, reassign the provided form.

The following keys exist on the form to show where in the lifecycle you are:
	submitted_once? - If the form has ever been submitted. Useful for not showing any errors on the first attempt to fill out a form.
	just_submitted? - If the form has just been submitted and no validation has happened since. Useful for things like
triggering a UI effect that should stop when the form is modified again.
	.changed? - If something about the form is different than it originally was. Note that in some cases this can yield a
false positive, specifically if a nested form is removed and then a new one is added with the exact same values.
	.touched_forms - A MapSet containing all keys in the form that have been modified. When submitting a form, only these keys are included in the parameters.


  
    
  
  Forms in the code interface


Throughout this documentation you will see forms created with AshPhoenix.Form.for_create/3 and other functions like it.
This is perfectly fine to do, however there is a way to use AshPhoenix.Form in a way that adds clarity to its usage
and makes it easier to find usage of each action. Code interfaces allow us to do this for standard action calls, i.e:
resources do
  resource MyApp.Accounts.User do
    define :register_with_password, args: [:email, :password]
    define :update_user, action: :update, args: [:email, :password]
  end
end
Adding the AshPhoenix extension to our domains and resources, like so:
use Ash.Domain,
  extensions: [AshPhoenix]
will cause another function to be generated for each definition, beginning with form_to_.
With this extension, the standard setup for forms looks something like this:
def render(assigns) do
  ~H"""
  <.form for={@form} phx-change="validate" phx-submit="submit">
    <.input field={@form[:email]} />
    <.input field={@form[:password]} />
    <.button type="submit" />
  </.form>
  """
end

def mount(_params, _session, socket) do
  # Here we call our new generated function to create the form
  {:ok, assign(socket, form: MyApp.Accounts.form_to_register_with_password())}
end

def handle_event(socket, "validate", %{"form" => params}) do
  form = AshPhoenix.Form.validate(socket.assigns.form, params)
  {:noreply, assign(socket, :form, form)}
end

def handle_event(socket, "submit", %{"form" => params}) do
  case AshPhoenix.Form.submit(socket.assigns.form, params: params) do
    {:ok, _user} ->
      socket =
        socket
        |> put_flash(:success, "User registered successfully")
        |> push_navigate(to: ~p"/")

      {:noreply, socket}

    {:error, form} ->
      socket =
        socket
        |> put_flash(:error, "Something went wrong")
        |> assign(:form, form)

      {:noreply, socket}
  end
end

  
    
  
  Working with related or embedded data


See the nested forms guide

  
    
  
  Working with compound types


Compound types, such as Ash.Money, will need some extra work to make it work.
For instance, when working with the Transfer type in AshDoubleEntry.Transfer, it will have the Ash.Money type for amount. When rendering the forms, you should do as follows:
<.input
    name={@form[:amount].name <> "[amount]"}
    id={@form[:amount].id <> "_amount"}
    label="Amount"
    value={if(@form[:amount].value, do: @form[:amount].value.amount)}
  />
  <.input
    type="select"
    name={@form[:amount].name <> "[currency]"}
    id={@form[:amount].id <> "_currency"}
    options={[:USD, :HKD, :EUR]}
    label="Currency"
    value={if(@form[:amount].value, do: @form[:amount].value.currency)}
  />
The above will allow the fields to be used by the AshPhoenix.Form when creating or updating a Transfer.
You can follow the same style with other compound types.

AshPhoenix.Form.Auto 
    



      
A tool to automatically generate available nested forms based on a resource and action.
To use this, specify forms: [auto?: true] when creating the form.
Keep in mind, you can always specify these manually when creating a form by simply specifying the forms option.
There are two things that this builds forms for:
	Attributes/arguments who's type is an embedded resource.
	Arguments that have a corresponding change manage_relationship(..) configured.

For more on relationships see the documentation for Ash.Changeset.manage_relationship/4.
When building forms, you can switch on the action type and/or resource of the form, in order to have different
fields depending on the form. For example, if you have a simple relationship called :comments with
on_match: :update and on_no_match: :create, there are two types of forms that can be in inputs_for(form, :comments).
In which case you may have something like this:
<%= for comment_form <- inputs_for(f, :comments) do %>
  <%= hidden_inputs_for(comment_form) %>
  <%= if comment_form.source.type == :create do %>
    <%= text_input comment_form, :text %>
    <%= text_input comment_form, :on_create_field %>
  <% else %>
    <%= text_input comment_form, :text %>
    <%= text_input comment_form, :on_update_field %>
  <% end %>

  <button phx-click="remove_form" phx-value-path="<%= comment_form.name %>">Remove Comment</button>
  <button phx-click="add_form" phx-value-path="<%= comment_form.name %>">Add Comment</button>
<% end %>
This also applies to adding forms of different types manually. For instance, if you had a "search" field
to allow them to search for a record (e.g in a liveview), and you had an on_lookup read action, you could
render a search form for that read action, and once they've selected a record, you could render the fields
to update that record (in the case of on_lookup: :relate_and_update configurations).

  
    
  
  Options


	:relationship_fetcher (term/0) - A two argument function that receives the parent data, the relationship to fetch.
The default simply fetches the relationship value, and if it isn't loaded, it uses [] or nil.

	:sparse_lists? (boolean/0) - Sets all list type forms to sparse?: true by default. Has no effect on forms derived for embedded resources. The default value is false.

	:include_non_map_types? (boolean/0) - Creates form for non map or array of map type inputs The default value is false.



  
    
  
  Special Considerations



  
    
  
  on_lookup: :relate_and_update


For on_lookup: :relate_and_update configurations, the "read" form for that relationship will use the appropriate read action.
However, you may also want to include the relevant fields for the update that would subsequently occur. To that end, a special
nested form called :_update is created, that uses an empty instance of that resource as the base of its changeset. This may require
some manual manipulation of that data before rendering the relevant form because it assumes all the default values. To solve for this,
if you are using liveview, you could actually look up the record using the input from the read action, and then use AshPhoenix.Form.update_form/3
to set that looked up record as the data of the _update form.

  
    
  
  Many to Many Relationships


In the case that a manage_change option points to a join relationship, that form is presented via a special nested form called
_join. So the first form in inputs_for(form, :relationship) would be for the destination, and then inside of that you could say
inputs_for(nested_form, :_join). The parameters are merged together during submission.

AshPhoenix.Form.WrappedValue 
    



      
A sentinal value used when editing a union that has non-map values

AshPhoenix.FormData.Error protocol
    



      
A protocol for allowing errors to be rendered into a form.
To implement, define a to_form_error/1 and return a single error or list of errors of the following shape:
{:field_name, message, replacements}
Replacements is a keyword list to allow for translations, by extracting out the constants like numbers from the message.

AshPhoenix.FilterForm 
    



      
A module to help you create complex forms that generate Ash filters.
# Create a FilterForm
filter_form = AshPhoenix.FilterForm.new(MyApp.Payroll.Employee)
FilterForm's comprise 2 concepts, predicates and groups. Predicates are the simple boolean
expressions you can use to build a query (name == "Joe"), and groups can be used to group
predicates and more groups together. Groups can apply and or or operators to its nested
components.
# Add a predicate to the root of the form (which is itself a group)
filter_form = AshPhoenix.add_predicate(filter_form, :some_field, :eq, "Some Value")

# Add a group and another predicate to that group
{filter_form, group_id} = AshPhoenix.add_group(filter_form, operator: :or, return_id?: true)
filter_form = AshPhoenix.add_predicate(filter_form, :another, :eq, "Other", to: group_id)
validate/1 is used to merge the submitted form params into the filter form, and one of the
provided filter functions to apply the filter as a query, or generate an expression map,
depending on your requirements:
filter_form = AshPhoenix.validate(socket.assigns.filter_form, params)

# Generate a query and pass it to the Domain
query = AshPhoenix.FilterForm.filter!(MyApp.Payroll.Employee, filter_form)
filtered_employees = MyApp.Payroll.read!(query)

# Or use one of the other filter functions
AshPhoenix.FilterForm.to_filter_expression(filter_form)
AshPhoenix.FilterForm.to_filter_map(filter_form)

  
    
  
  LiveView Example


You can build a form and handle adding and removing nested groups and predicates with the following:
alias MyApp.Payroll.Employee

@impl true
def render(assigns) do
  ~H"""
  <.simple_form
    :let={filter_form}
    for={@filter_form}
    phx-change="filter_validate"
    phx-submit="filter_submit"
  >
    <.filter_form_component component={filter_form} />
    <:actions>
      <.button>Submit</.button>
    </:actions>
  </.simple_form>
  <.table id="employees" rows={@employees}>
    <:col :let={employee} label="Payroll ID"><%= employee.employee_id %></:col>
    <:col :let={employee} label="Name"><%= employee.name %></:col>
    <:col :let={employee} label="Position"><%= employee.position %></:col>
  </.table>
  """
end

attr :component, :map, required: true, doc: "Could be a FilterForm (group) or a Predicate"

defp filter_form_component(%{component: %{source: %AshPhoenix.FilterForm{}}} = assigns) do
  ~H"""
  <div class="border-gray-50 border-8 p-4 rounded-xl mt-4">
    <div class="flex flex-row justify-between">
      <div class="flex flex-row gap-2 items-center">Filter</div>
      <div class="flex flex-row gap-2 items-center">
        <.input type="select" field={@component[:operator]} options={["and", "or"]} />
        <.button phx-click="add_filter_group" phx-value-component-id={@component.source.id} type="button">
          Add Group
        </.button>
        <.button
          phx-click="add_filter_predicate"
          phx-value-component-id={@component.source.id}
          type="button"
        >
          Add Predicate
        </.button>
        <.button
          phx-click="remove_filter_component"
          phx-value-component-id={@component.source.id}
          type="button"
        >
          Remove Group
        </.button>
      </div>
    </div>
    <.inputs_for :let={component} field={@component[:components]}>
      <.filter_form_component component={component} />
    </.inputs_for>
  </div>
  """
end

defp filter_form_component(
       %{component: %{source: %AshPhoenix.FilterForm.Predicate{}}} = assigns
     ) do
  ~H"""
  <div class="flex flex-row gap-2 mt-4">
    <.input
      type="select"
      options={AshPhoenix.FilterForm.fields(Employee)}
      field={@component[:field]}
    />
    <.input
      type="select"
      options={AshPhoenix.FilterForm.predicates(Employee)}
      field={@component[:operator]}
    />
    <.input field={@component[:value]} />
    <.button
      phx-click="remove_filter_component"
      phx-value-component-id={@component.source.id}
      type="button"
    >
      Remove
    </.button>
  </div>
  """
end

@impl true
def mount(_params, _session, socket) do
  socket =
    socket
    |> assign(:filter_form, AshPhoenix.FilterForm.new(Employee))
    |> assign(:employees, Employee.read_all!())

  {:ok, socket}
end

@impl true
def handle_event("filter_validate", %{"filter" => params}, socket) do
  {:noreply,
   assign(socket,
     filter_form: AshPhoenix.FilterForm.validate(socket.assigns.filter_form, params)
   )}
end

def handle_event("filter_submit", %{"filter" => params}, socket) do
  filter_form = AshPhoenix.FilterForm.validate(socket.assigns.filter_form, params)

  case AshPhoenix.FilterForm.filter(Employee, filter_form) do
    {:ok, query} ->
      {:noreply,
       socket
       |> assign(:employees, Employee.read_all!(query: query))
       |> assign(:filter_form, filter_form)}

    {:error, filter_form} ->
      {:noreply, assign(socket, filter_form: filter_form)}
  end
end

def handle_event("remove_filter_component", %{"component-id" => component_id}, socket) do
  {:noreply,
   assign(socket,
     filter_form:
       AshPhoenix.FilterForm.remove_component(socket.assigns.filter_form, component_id)
   )}
end

def handle_event("add_filter_group", %{"component-id" => component_id}, socket) do
  {:noreply,
   assign(socket,
     filter_form: AshPhoenix.FilterForm.add_group(socket.assigns.filter_form, to: component_id)
   )}
end

def handle_event("add_filter_predicate", %{"component-id" => component_id}, socket) do
  {:noreply,
   assign(socket,
     filter_form:
       AshPhoenix.FilterForm.add_predicate(socket.assigns.filter_form, :name, :contains, nil,
         to: component_id
       )
   )}
end

AshPhoenix.FilterForm.Arguments 
    



      
Represents the arguments to a calculation being filtered on

AshPhoenix.FilterForm.Predicate 
    



      
Represents an individual predicate appearing in a filter form.
Predicates are grouped up in an AshPhoenix.FilterForm to create boolean
filter statements.

AshPhoenix.Form.InvalidPath exception
    



      
Raised when an invalid path is used to find, update or remove a form

    AshPhoenix.Form.NoActionConfigured - ash_phoenix v2.1.14
    
    

    


  
  

    
AshPhoenix.Form.NoActionConfigured exception
    



      
Raised when a form action should happen but no action of the appropriate type has been configured

    AshPhoenix.Form.NoDataLoaded - ash_phoenix v2.1.14
    
    

    


  
  

    
AshPhoenix.Form.NoDataLoaded exception
    



      
Raised when a data needed to be used but the required data was not loaded

      


      
        Summary


  
    Functions
  


    
      
        
  
    
    AshPhoenix.Form.NoFormConfigured - ash_phoenix v2.1.14
    
    

    


  
  

    
AshPhoenix.Form.NoFormConfigured exception
    



      
Raised when attempting to refer to a form but no nested form with that name was configured.

    AshPhoenix.Form.NoResourceConfigured - ash_phoenix v2.1.14
    
    

    


  
  

    
AshPhoenix.Form.NoResourceConfigured exception
    



      
Raised when a form needed to be constructed but the resource for that form could not be determined

mix ash_phoenix.gen.html 
    



      
This task renders .ex and .heex templates and copies them to specified directories.

  
    
  
  Positional Arguments


	domain - The domain (e.g. "Shop").
	resource - The resource (e.g. "Product").


  
    
  
  Options


	--resource-plural - The plural resource name (e.g. "products")

mix ash_phoenix.gen.html MyApp.Shop MyApp.Shop.Product --plural-name products

mix ash_phoenix.gen.live 
    



      
Generates liveviews for a given domain and resource.
The domain and resource must already exist, this task does not define them.

  
    
  
  Example


mix ash_phoenix.gen.live --domain ExistingDomainName --resource ExistingResourceName --resourceplural ExistingResourceNames


  
    
  
  Options


	--domain   - Existing domain
	--resource - Existing resource
	--resourceplural - Plural resource name

